***AJUSTE***
rename categ categcbo
rename CAT8 cat

***RECORTES***

*Recorte geográfico (rurais do Norte)

generate rec_geo = 1 
replace rec_geo = 0 if (uf>=11 & uf<=16) & (V1022==2)
label variable rec_geo "Recorte geográfico (tirando rurais do Norte)"

generate rec_idade = 0 
replace rec_idade = 1 if (V2009>=18 & V2009<=24)
label variable rec_idade "Recorte etário (selecionando 18-24 anos)"

generate rec_pos = 0 
replace rec_pos = 1 if (V2005>=4 & V2005<=6)
label variable rec_pos "Recorte posição no domicílio (selecionando apenas filhos)"

generate rec_mod = 0 
replace rec_mod = 1 if (rec_geo==1 & rec_idade==1 & rec_pos==1)
label variable rec_mod "Recorte total para modelos"

***PESO***
generate peso = V1032
label variable peso "peso amostral"




***VARIÁVEIS INDEPENDENTES

*Versão PNAD
generate versao = 1
label define VERSAO 0 "PNAD antiga" 1 "PNADc"
label values versao VERSAO
label variable versao "VERSAO" 

*Ano
generate ano = .
replace ano = Ano
label variable ano "Ano da pesquisa" 


*Idade
generate idade = .
replace idade = V2009
label variable idade "Idade, em anos" 

*Sexo
generate sexo = .
replace sexo = 0 if V2007==1
replace sexo = 1 if V2007==2
label define SEXO 0 "Homens" 1 "Mulheres"
label values sexo SEXO
label variable sexo "Sexo" 

*Cor ou raça
generate cor = .
replace cor = 0 if (V2010==1 | V2010==3)
replace cor = 1 if (V2010==2 | V2010==4 | V2010==5)
label define COR 0 "Branco" 1 "Não-branco"
label values cor COR
label variable cor "Cor ou raça" 

*Região geográfica
generate regiao = .
replace regiao = 1 if (uf>=11 & uf<=17)
replace regiao = 2 if (uf>=21 & uf<=29)
replace regiao = 3 if (uf>=31 & uf<=35)
replace regiao = 4 if (uf>=41 & uf<=43)
replace regiao = 5 if (uf>=50 & uf<=53)
label define REGIAO 1 "Norte" 2 "Nordeste" 3 "Sudeste" 4 "Sul" 5 "Centro-Oeste"
label values regiao REGIAO
label variable regiao "Região geográfica" 

*Local
generate local = .
replace local = 0 if V1022==1
replace local = 1 if V1022==2
label define LOCAL 0 "Urbano" 1 "Rural"
label values local LOCAL
label variable local "Localização geográfica"

*Família monoparental
generate conj = 1 if V2005==2 | V2005==3
by Ano Trimestre upa  V1008  V1014, sort : egen conj_agg = max(conj)

generate tipo_fam = .
replace tipo_fam = 0 if conj_agg==.
replace tipo_fam = 1 if conj_agg==1
label define FAMILIA 0 "Sem cônjuge" 1 "Com Cônjuge"
label values tipo_fam FAMILIA
label variable tipo_fam "Tipo de família"

*Escolaridade do chefe de domicílio
generate esc_pd8 = .
replace esc_pd8 = VD3005 if VD3005>=0 & VD3005<=8
replace esc_pd8 = VD3005-1 if VD3005>=9 & VD3005<=16

generate esc_chefe = .
replace esc_chefe = esc_pd8 if V2005==1
label variable esc_chefe "Anos de estudo (quando for chefe de domicílio)"

by Ano Trimestre upa  V1008  V1014, sort : egen esc_chefe_agg = max(esc_chefe)
label variable esc_chefe_agg "Anos de estudo do chefe do domicílio onde o entrevistado reside"

*Número de irmãos
generate filhos = 1 if V2005==4 | V2005==5 | V2005==6
by Ano Trimestre upa  V1008  V1014, sort : egen filhos_agg = sum(filhos)

generate filhos_agg2 = .
replace filhos_agg2 = filhos_agg-1 if (V2005==4 | V2005==5 | V2005==6)
label variable filhos_agg2 "Número de irmãos conviventes (somente para filhos)"

*CAT do responsável pelo domicílio
generate cat_chefe = .
replace cat_chefe = cat if V2005==1
label variable cat_chefe "CAT (quando for chefe de domicílio)"

by Ano Trimestre upa  V1008  V1014, sort : egen cat_chefe_agg = max(cat_chefe)
label variable cat_chefe_agg "CAT do chefe do domicílio onde o entrevistado reside"

label define CATC 1 "Grandes empregadores" 2 "Profissionais de nível superior" 3 "Pequenos empregadores" 4 "Ocupações médias" 5 "Trabalhadores do terciário especializado" 6 "Trabalhadores do secundário" 7 "Trabalhadores do terciário não-especializado" 8 "Trabalhadores agrícolas"
label values cat_chefe_agg CATC

*Renda domiciliar per capita
generate renda_dom_pcta = VD5008*co2
label variable renda_dom_pcta "Renda domiciliar per capita (que não pensionistas, empregados domésticos, parentes dos empregados domésticos)"



***VARIÁVEIS DEPENDENTES***

*Terminou o Ensino Médio?
generate ens_medio = . 
replace ens_medio = 0 if (V3003>=1 & V3003<=6) | (V3009>=1 & V3009<=6) | ((V3009>=7 & V3009<=9) & (V3014==2)) | ((V3009>=7 & V3009<=9) & (V3014==.) & (VD3005>=0 & VD3005<=11)) | ((V3003==. & V3009==. & VD3005==0))
replace ens_medio = 1 if (V3003==7 | V3003==8 | V3003==9) | (V3009==10 | V3009==11 | V3009==12) | ((V3009==7 | V3009==8 | V3009==9) & (V3014==1)) | ((V3009==7 | V3009==8 | V3009==9) & (V3014==.) & (VD3005>=12 & VD3005<=16))
label define DIC 0 "Não" 1 "Sim"
label values ens_medio DIC
label variable ens_medio "Terminou o Ensino Médio?"


*Ingressou no Ensino Superior?
generate ens_sup = . 
replace ens_sup = 0 if (V3003>=1 & V3003<=6) | (V3009>=1 & V3009<=9) | ((V3003==. & V3009==. & VD3005==0))
replace ens_sup = 1 if (V3003==7 | V3003==8 | V3003==9) | (V3009==10 | V3009==11 | V3009==12) 
label values ens_sup DIC
label variable ens_sup "Ingressou no Ensino Superior?"


*Frequenta nesse momento o Ensino Superior?
generate ens_sup_a = 0 
replace ens_sup_a = 1 if (V3003>=7 & V3003<=9) 
label values ens_sup_a DIC
label variable ens_sup_a "Frequenta nesse momento o Ensino Superior?"



